Problem:
Score : 300 points

Problem Statement

You are given a grid of squares with H horizontal rows and W vertical columns, where each square is painted white or black.
HW characters from A_{11} to A_{HW} represent the colors of the squares.
A_{ij} is # if the square at the i-th row from the top and the j-th column from the left is black, and A_{ij} is . if that square is white.

We will repeatedly perform the following operation until all the squares are black:

Every white square that shares a side with a black square, becomes black.

Find the number of operations that will be performed.
The initial grid has at least one black square.

Constraints

1 \leq H,W \leq 1000

A_{ij} is # or ..

The given grid has at least one black square.

Input

Input is given from Standard Input in the following format:

H W
A_{11}A_{12}...A_{1W}
:
A_{H1}A_{H2}...A_{HW}

Output

Print the number of operations that will be performed.

Sample Input 1

3 3
...
.#.
...

Sample Output 1

2

After one operation, all but the corners of the grid will be black. After one more operation, all the squares will be black.

Sample Input 2

6 6
..#..#
......
#..#..
......
.#....
....#.

Sample Output 2

3

Solution in OCaml (4.02.3) (judge verdict: Accepted)


module PQ = Set.Make(struct
  type t = int * int * int
  let compare (d0,x0,y0) (d1,x1,y1) =
    if d0 <> d1 then d0-d1 else if x0 <> x1 then x0-x1 else y0-y1
end)

let () =
  Scanf.scanf "%d %d" @@ fun h w ->
  let a = Array.init h (fun _ -> Scanf.scanf " %s" (fun s -> s)) in
  let a = Array.mapi (fun i s -> Array.init w (fun j -> i, j, s.[j])) a in

  let pq0 =
    Array.fold_left (Array.fold_left
      (fun pq (i,j,c) -> if c = '#' then PQ.add (0,i,j) pq else pq)) PQ.empty a
  in

  let c = Array.init h (fun _ -> Array.make w max_int) in
  PQ.iter (fun (_,x,y) -> c.(x).(y) <- 0) pq0;

  let rec loop pq =
    if PQ.is_empty pq then () else
    let (d,x,y) = PQ.min_elt pq in
    let pq = PQ.remove (d,x,y) pq in
    if d > c.(x).(y) then loop pq else
    [(1,0); (0,1); (-1,0); (0,-1)] |> List.fold_left (fun pq (dx,dy) ->
      let x', y' = x+dx, y+dy in
      if x' < 0 || h <= x' || y' < 0 || w <= y' || d+1 >= c.(x').(y') then pq
      else begin
        c.(x').(y') <- d+1;
        PQ.add (d+1,x',y') pq
      end
    ) pq |> loop
  in
  loop pq0;

  Array.fold_left (Array.fold_left max) 0 c
  |> Printf.printf "%d\n"